# $Id: PKGBUILD 86818 2013-03-22 12:06:39Z seblu $
# Maintainer: Sébastien Luttringer
# Contributor: SpepS <dreamspepser at yahoo dot it>
# Contributor: Laszlo Papp <djszapi at archlinux us>
# Contributor: Donald Ephraim Curtis <dcurtis@gmail.com>

pkgbase=penlight
pkgname=('lua-penlight')
pkgver=1.1.0
pkgrel=1
pkgdesc='Lua libraries focusing on input data handling'
arch=('any')
url='https://github.com/stevedonovan/penlight'
license=('custom')
makedepends=('lua')
source=("$pkgname-$pkgver.tar.gz::https://github.com/stevedonovan/penlight/archive/$pkgver.tar.gz")
md5sums=('835ab76e6330d9f584246c8a5113e4dc')

package_lua-penlight() {
  depends=('lua' 'lua-filesystem')

  cd Penlight-$pkgver
  install -dm755 "$pkgdir/usr/share/lua/5.2/pl/platf"
  install -m644 lua/pl/*.lua "$pkgdir/usr/share/lua/5.2/pl"
  install -m644 lua/pl/platf/*.lua "$pkgdir/usr/share/lua/5.2/pl/platf"
  # copy docs
  install -dm755 "$pkgdir/usr/share/docs/$pkgname"
  install -m644 CHANGES.txt README.txt docs/manual/* "$pkgdir/usr/share/docs/$pkgname"
  # copy license
  install -Dm644 LICENCE.txt "$pkgdir/usr/share/licenses/$pkgname/LICENSE"
}

# vim:set ts=2 sw=2 et:
